

# Experiencia de Laboratorio 03

Pensamiento Computacional y Programación  
Colegio La Girouette

---

Profesores: Darío Creado  
Experiencia : LAB03 - Dado de LEDs  
Fecha de Trabajo: 13 de Septiembre 2022

---

## 1 Descripción de la Experiencia y Circuito Esquemático

Este proyecto utiliza lo que acabamos de aprender sobre números aleatorios (explicado en la charla introductoria) para crear un dado electrónico con seis LEDs y un botón. Cada vez que se pulsa el botón, el LED "rueda" durante un tiempo antes de quedarse en un valor, para luego comunicarlo mediante destellos.

El circuito esquemático se muestra en la figura. Cada LED se maneja con una salida digital independiente mediante una resistencia limitadora de corriente. Aparte de esto lo único que hace falta es el pulsador y la correspondiente resistencia asociada de puesta a tierra. Aunque los dados convencionales tienen seis puntos, seguimos necesitando siete LEDs para conseguir la disposición normal de un punto en el centro para mostrar los números impares.<sup>1</sup>

Este programa es sencillo, pero aún así tiene pequeños detalles que hacen que el dado se comporte de manera similar a un dado real. Por ejemplo, cuando el dado "rueda", el número cambia pero va disminuyendo gradualmente. Además, la cantidad de tiempo que "rueda" el dado también es aleatoria.




Figure 1: Circuito Esquemático Proyecto Dado Aleatorio

---

<sup>1</sup>Simon Monk. Thirty Projects with Arduino

## 2 Consideraciones

Mucho cuidado con el manejo de *arrays* o arreglos, estos guardarán la puntuación obtenida en el lanzamiento aleatorio. También para este laboratorio se introduce el manejo de la instrucción **for** en el control de flujo del código, esta instrucción itera sobre otra estructura de forma de revisar cada componente en forma secuencial. Encontrarán al igual que en el laboratorio anterior, diversas funciones declaradas al final del código.

Consideren el siguiente resumen del código fuente :

Tenemos siete LEDs que inicializar en la función **setup**, por lo que vale la pena ponerlos en un arreglo o matriz (**dicePatterns**) y crear un bucle (**loop**) que vaya recorriendo la matriz para inicializar cada pin. También tenemos una llamada a **randomSeed** en el bloque **setup**, que si no estuviera allí, cada vez que hiciéramos un reset de la placa terminaríamos siempre con la misma secuencia de lanzamientos del dado. Como experimento, si lo deseas, prueba a convertir esta línea en el comentario colocando una `//` al principio de la misma para comprobarlo. El *array* **dicePatterns** determina que LEDs deben estar encendidos o apagados en una jugada determinada. De forma que, cada elemento de tiro de la matriz, es en realidad en sí mismo otra matriz de siete elementos, pudiendo ser cada uno de ellos HIGH o LOW (1 ó 0). Cuando vamos a mostrar el resultado concreto de haber tirado los dados, podemos simplemente recorrer la matriz del tiro, ajustando cada LED en consecuencia.

## 3 Materiales

1. Placa Arduino UNO o Similar.
2. LED rojo de 5mm ( $\times 7$ ).
3. Resistencia 270  $\Omega$  0.5 W ( $\times 7$ ).
4. Resistencia 100K  $\Omega$  0.5 W.
5. Pulsador o *switch* para conmutación.

## 4 Código Fuente

---

```
1  int ledPins[7] = {2,3,4,5,6,7,8};
2  int dicePatterns[7][7] = {
3      {0,0,0,0,0,0,1}, // 1
4      {0,0,1,1,0,0,0}, // 2
5      {0,0,1,1,0,0,1}, // 3
6      {1,0,1,1,0,1,0}, // 4
7      {1,0,1,1,0,1,1}, // 5
8      {1,1,1,1,1,1,0}, // 6
9      {0,0,0,0,0,0,0} // En blanco
10 };
11
12 int switchPin = 9;
13 int blank = 6;
14
15 void setup()
16 {
17     for (int i = 0; i < 7; i++)
18     {
19         pinMode(ledPins[i],OUTPUT);
20         digitalWrite(ledPins[i],LOW);
21     }
22     randomSeed(analogRead(0));
23 }
24
25 void loop()
26 {
27     if (digitalRead(switchPin))
28     {
29         rollTheDice();
30     }
31     delay(100);
32 }
33
34 void rollTheDice()
35 {
36     int result = 0;
37     int lenghtOfRoll = random(15,25);
38     for (int i = 0; i < lenghtOfRoll; i++)
39     {
40         result = random(0,6);
41         show(result);
42         delay(50+i*10);
43     }
44     for (int j = 0; j < 3; j++)
45     {
46         show(blank);
47         delay(500);
48         show(result);
49         delay(500);
50     }
51 }
52
53 void show(int result)
54 {
55     for (int i = 0; i < 7; i++)
56     {
57         digitalWrite(ledPins[i],dicePatterns[result][i]);
58     }
59 }
```

---